

This guide details a few important functions of FRANZ LISP and COMMON LISP.

Note, that all predefined (i.e., built-in) function and symbol names in FRANZ LISP are in lower case. That includes `nil` and `t`. In COMMON LISP, however, they are defined in upper case. The expression reader converts all symbols to upper case, so that you may use either upper or lower case in your programs.

Intrinsic Functions

These are the functions which come predefined, and are available at sign-on to LISP. This is not a complete list. Functions that are specific to FRANZ LISP are marked with [F], whereas functions that are specific to COMMON LISP are marked with [C].

1 S-expression manipulation

<code>(car arg)</code>	Returns the first element in list <code>arg</code> , or the left pointer of an s-expression (cons node).
<code>(cdr arg)</code>	Return the rest of list <code>arg</code> , or the right pointer of an s-expression (cons node).
<code>(cons a1 a2)</code>	Allocate and return a cons node that points to <code>a1</code> and <code>a2</code> .
<code>(list a1 a2 ...an)</code>	Return a list of all the arguments.
<code>(append a1 a2)</code>	Return the merge of the two lists <code>a1</code> and <code>a2</code> .
<code>(reverse arg)</code>	Reverses the list <code>arg</code> .
<code>(subst new old sexp)</code>	Return a copy of <code>sexp</code> with all <code>old</code> 's converted to <code>new</code> 's.

2 S-expression predicates

<code>(null arg)</code>	Return <code>t</code> if <code>arg</code> is <code>nil</code> ; <code>nil</code> otherwise.
<code>(atom arg)</code>	Return <code>nil</code> if <code>arg</code> is a cons node; <code>t</code> otherwise.
<code>(eq a1 a2)</code>	Return <code>t</code> if <code>a1</code> and <code>a2</code> are the same pointer; <code>nil</code> otherwise.
<code>(equal a1 a2)</code>	Return <code>t</code> if <code>a1</code> and <code>a2</code> have the same structure; <code>nil</code> otherwise.
<code>(member a list)</code>	Return the sublist of <code>list</code> beginning with the first occurrence of <code>a</code> ; <code>nil</code> if not in <code>list</code> .

3 Logical functions

<code>(or a1 a2 ...an)</code>	Return <code>t</code> if any argument is non- <code>nil</code> ; <code>nil</code> otherwise.
<code>(and a1 a2 ...an)</code>	Return <code>t</code> if all arguments are non- <code>nil</code> ; <code>nil</code> otherwise.
<code>(not arg)</code>	If <code>arg</code> is <code>nil</code> , returns <code>t</code> ; else returns <code>nil</code> .

4 Control and program functions

<code>(quote a)</code> or <code>'a</code>	Return a unevaluated.
<code>(cond</code> <code>(exp1 a1 a2 ...an)</code> <code>(exp2 b1 b2 ...bm)</code> <code>...</code> <code>(expn z1 z2 ...zk))</code>	Evaluates all <code>exp</code> until one is non- <code>nil</code> , then evaluates all expressions that follow and returns the result of the last.
<code>(prog (v1 v2 ...vn)</code> <code>label1 (s-expr1)</code> <code>label2 (s-expr2)</code> <code>...</code> <code>labelm (s-exprm))</code>	Acts like a sequential program with local variables <code>v1</code> to <code>vn</code> . Labels are optional for each line. If no return is encountered and control “drops off the end,” then <code>nil</code> is returned.
<code>(go arg)</code>	In a <code>prog</code> , branch control to <code>s-expr_i</code> preceded by label <code>arg</code> .
<code>(return arg)</code>	Only valid in a <code>prog</code> . Terminate <code>prog</code> and return the value of <code>arg</code> .

5 I/O Functions

<code>(read)</code>	Return as a value the next <code>s-expression</code> typed as input to the terminal.
<code>(readc) [F]</code>	Return next character as an atom.
<code>(read-char) [C]</code>	
<code>(print arg)</code>	Print <code>arg</code> to output.
<code>(patom arg) [F]</code>	Print <code>arg</code> to output. Differs from <code>print</code> in that the output is a bit more readable, e.g., strings are not enclosed in quotes.
<code>(terpr) [F]</code>	Print a newline character.
<code>(terpri) [C]</code>	
<code>(pp arg) [F]</code>	Print the definition of the symbol <code>arg</code> . If <code>arg</code> is a function, then its binding is pretty printed. <code>pp</code> can take multiple arguments.
<code>(symbol-function atom) [C]</code>	Print the function definition of the function bound to <code>atom</code> .
<code>(cprintf format args) [F]</code>	Analogous to the C language version.
<code>(load filename)</code>	Read in the file <code>filename</code> and evaluates all expressions in it, including function definitions. Values of expressions are not displayed.
<code>(include filename) [F]</code>	Same as <code>load</code> , except that the argument is not evaluated, so it shouldn't be quoted.

6 Arithmetic Functions

<code>(numberp n)</code>	Return <code>t</code> if <code>n</code> is a number; <code>nil</code> otherwise.
<code>(zerop n)</code>	Return <code>t</code> if <code>n</code> is zero; <code>nil</code> otherwise.
<code>(> n1 n2)</code> or <code>(greaterp n1 n2)</code> [F]	Return <code>t</code> if <code>n1</code> is greater than <code>n2</code> .
<code>(< n1 n2)</code> or <code>(lessp n1 n2)</code> [F]	Return <code>t</code> if <code>n1</code> is less than <code>n2</code> .
<code>(+ n1 n2 ...nn)</code> or <code>(plus n1 n2 ...nn)</code> [F]	Return the sum of all arguments.
<code>(* n1 n2 ...nn)</code> or <code>(times n1 n2 ...nn)</code> [F]	Return the product of all arguments.
<code>(- n1 n2)</code> or <code>(difference n1 n2)</code> [F]	Return the quantity <code>n1</code> minus <code>n2</code> .
<code>(/ n1 n2)</code> or <code>(quotient n1 n2)</code> [F]	Return <code>n1</code> divided by <code>n2</code> .
<code>(1+ x)</code> or <code>(add1 x)</code> [F]	Return <code>x + 1</code> .
<code>(1- x)</code> or <code>(sub1 x)</code> [F]	Return <code>x - 1</code> .

7 Function Definition and Value Assignment

<code>(setq x y)</code>	Set <code>x</code> to the value of <code>y</code> . <code>setq</code> allows any number of <code>x-y</code> pairs.
<code>(set x y)</code>	Like <code>setq</code> , except that <code>x</code> is evaluated to get an atom.
<code>(lambda arglist body)</code>	Return a nameless function with argument list <code>arglist</code> and body <code>body</code> . The body should be a list of expression, with the value of the last one being the value of the function.
<code>(defun name arglist body)</code>	Define a function with name <code>name</code> . Equivalent to <code>(setq name (lambda arglist body))</code> .
<code>(defun name fexpr (arg) body)</code> [F]	Define a function where upon invocation, the argument list is not evaluated, but passed instead as the binding of the only parameter <code>arg</code> (not a built-in function in COMMON LISP).
<code>(defun.fexpr name (arg) body)</code> [C]	

8 Atom Manipulation

<code>(gensym arg)</code>	Create atom with name <code>argnnnnn</code> , where <code>arg</code> is an atom and <code>nnnnn</code> is the number of times <code>gensym</code> has been called. <code>(gensym 'x)</code> returns <code>x00000</code> .
<code>(putprop atom value label) [F]</code>	Set the value of the property <code>label</code> of atom's property list to <code>value</code> .
<code>(setf (get atom label) value) [C]</code>	
<code>(get atom label)</code>	Return the value of property <code>label</code> on the property list of atom.
<code>(remprop atom label)</code>	Remove the property <code>label</code> from the property list of atom.
<code>(plist atom) [F]</code>	Return the property list of atom in the form <code>(p1 v1 p2 v2 ... pn vn)</code> , where <code>pi</code> is a property with value <code>vi</code> .
<code>(symbol-plist atom) [C]</code>	
<code>(setplist atom list) [F]</code>	Set the property list of atom to <code>list</code> , which must be a list of the form <code>(p1 v1 p2 v2 ... pn vn)</code> .
<code>(setf (symbol-plist atom) list) [C]</code>	

9 Other useful functions

<code>(implode list) [F]</code>	Return the atom created by concatenating the first character of all the atoms in <code>list</code> .
<code>(explode atom) [F]</code>	Return the list of the atom's characters (reverse of <code>implode</code>).
<code>(concat-symbols symbol1 symbol2) [C]</code>	Return a symbol with a name that is a concatenation of the names of <code>symbol1</code> and <code>symbol2</code> (not a built-in function).
<code>(nth number list)</code>	Return the element of <code>list</code> with index number, assuming zero-based indexing.
<code>(nthcdr number list)</code>	Return the result of applying <code>cdr</code> to the <code>list</code> number times.
<code>(length list)</code>	Return the number of elements in the top level of <code>list</code> .
<code>(trace fname ...)</code>	Turn on function tracing for function <code>fname</code> . Trace can take multiple arguments. A list of all functions being traced is returned.
<code>(untrace fname ...)</code>	Turn off tracing for the named functions.
<code>(help)</code>	On-line help. <code>help</code> can take an argument, e.g., a function name.
<code>(vi filename) [F]</code>	Invoke the vi editor on the file <code>filename</code> .
<code>(vil filename) [F]</code>	Same as <code>vi</code> , except that <code>load</code> is executed on the file after vi is exited.

10 More hints

In order to invoke the emacs editor instead of vi in FRANZ LISP, use the following functions `em` and `eml` instead of `vi` and `vil`, respectively:

```
(defun em fexpr (x) (exvi 'emacs x nil))
(defun eml fexpr (x) (exvi 'emacs x t))
```

To save the output of your program into a file, you can use the UNIX command `script`. To use it, first execute `script`, then produce the output you want to save, and, finally, execute `exit` in the shell. At that point, the output will be in the file `typescript`, which you can then edit and print.

The COMMON LISP functions `defun.fexpr` and `concat-symbols` that were mentioned above are not a built-in functions. Their definition follows. The functions `trace.fexpr` and `untrace.fexpr` are used to trace and untrace `fexpr` functions.

```
(defmacro defun.fexpr (funname varlist &rest expr)
  (let ((ffunname (concat-symbols funname '.fexpr)))
    `(progn
      (defun ,ffunname ,varlist ,@expr)
      (defmacro ,funname (&body args)
        (list ',ffunname (list 'quote args))))))

(defun fexprlist (el)
  (if (null el) nil
      (cons (concat-symbols (car el) '.fexpr)
            (fexprlist (cdr el)))))

(defmacro trace.fexpr (&body funname)
  (cons 'trace (fexprlist funname)))

(defmacro untrace.fexpr (&body funname)
  (cons 'untrace (fexprlist funname)))

(defun concat-symbols (sym1 sym2)
  (intern (concatenate 'string (string sym1) (string sym2))))
```